Includes barcode for easy software entry.

HEWLETT-PACKARD

HP-41

USERS' LIBRARY SOLUTIONS

Antennas



INTRODUCTION

This HP-41C Solutions book was written to help you get the most from your calculator. The programs were chosen to provide useful calculations for many of the common problems encountered.

They will provide you with immediate capabilities in your everyday calculations and you will find them useful as guides to programming techniques for writing your own customized software. The comments on each program listing describe the approach used to reach the solution and help you follow the programmer's logic as you become and expert on your HP calculator.

KEYING A PROGRAM INTO THE HP-41C

There are several things that you should keep in mind while you are keying in programs from the program listings provided in this book. The output from the HP 82143A printer provides a convenient way of listing and an easily understood method of keying in programs without showing every keystroke. This type of output is what appears in this handbook. Once you understand the procedure for keying programs in from the printed listings, you will find this method simple and fast. Here is the procedure:

1. At the end of each program listing is a listing of status information required to properly execute that program. Included is the SIZE allocation required. Before you begin keying in the program, press [XEQ] [ALPHA] SIZE [ALPHA] and specify the allocation (three digits; e.g., 10 should be specified as 010).

   Also included in the status information is the display format and status of flags important to the program. To ensure proper execution, check to see that the display status of the HP-41C is set as specified and check to see that all applicable flags are set or clear as specified.

2. Set the HP-41C to PRGM mode (press the [PRGM] key) and press ■ [GTO] [•] [•] to prepare the calculator for the new program.

3. Begin keying in the program. Following is a list of hints that will help you when you key in your programs from the program listings in this handbook.

   a. When you see " (quote marks) around a character or group of characters in the program listing, those characters are ALPHA. To key them in, simply press [ALPHA], key in the characters, then press [ALPHA] again. So "SAMPLE" would be keyed in as [ALPHA] "SAMPLE" [ALPHA].

   b. The diamond in front of each LBL instruction is only a visual aid to help you locate labels in the program listings. When you key in a program, ignore the diamond.

   c. The printer indication of divide sign is /. When you see / in the program listing, press [÷].

   d. The printer indication of the multiply sign is ⁎ . When you see ⁎ in the program listing, press [×].

   e. The ⊢ character in the program listing is an indication of the [APPEND] function. When you see ⊢, press ■ [APPEND] in ALPHA mode (press ■ and the K key).

   f. All operations requiring register addresses accept those addresses in these forms:

      nn (a two-digit number)
      IND nn (INDIRECT: ■ , followed fy a two-digit number)
      X, Y, Z, T, or L (a STACK address: [•] followed by X, Y, Z, T, or L)
      IND X, Y, Z, T or L (INDIRECT stack: ■ [•] followed by X, Y, Z, T, or L)

      Indirect addresses are specified by pressing ■ and then the indirect address. Stack addresses are specified by pressing [•] followed by X, Y, Z, T, or L. Indirect stack addresses are specified by pressing ■ [•] and X, Y, Z, T, or L.

Printer Listing	Keystrokes	Display
01♦LBL "SAMPLE"	■ [LBL] [ALPHA] SAMPLE [ALPHA]	*01 LBL^{T}SAMPLE*
02 "THIS IS A "	[ALPHA] *THIS IS A* [ALPHA]	*02^{T}THIS IS A*
03 "⊢SAMPLE"	[ALPHA] ■ [APPEND] SAMPLE	*03^{T} ⊢ SAMPLE*
04 AVIEW	■ AVIEW [ALPHA]	*04 AVIEW*
05 6	6	*05 6*
06 ENTER↑	[ENTER↑]	*06 ENTER ↑*
07 -2	2 [CHS]	*07 -2*
08 /	[÷]	*08 /*
09 ABS	[XEQ] [ALPHA] ABS [ALPHA]	*09 ABS*
10 STO IND L	[STO] ■ [•] L	*10 STO IND L*
11 "R3="	[ALPHA] R3= ■ [ARCL] 03	*11^{T}R3=*
12 ARCL 03	■ [AVIEW]	*12 ARCL 03*
13 AVIEW	[ALPHA]	*13 AVIEW*
14 RTN	■ [RTN]	*14 RTN*

TABLE OF CONTENTS

LOADED VERTICAL ANTENNAS

Let A = Height above ground in inches

B = Length of lower section in inches

C = Length of upper section in inches

a = Average thickness in inches

F = Frequency in MHz

$$\lambda = V_c/F = 3\text{x}10^4/2.54F$$

$$Z_0 = 138 \log \left[(A + B + C)/a\right]$$

Electrical length of each section

$$B° = B \cdot 360/\lambda = B \cdot 360 \cdot 2.54F/3\text{x}10^4 = B \cdot F/32.81 \text{ degrees}$$

$$C° = C \cdot F/32.81 \text{ degrees}$$

Required Reactance

$$jX_L = jZ_0 \text{ (cot } C° - \tan B°)$$

Inductance $L = jX_L/2\pi F = \mu Hy$

$$R_{rad} = \left[(C°\cos B°)/2 + B° \ (1 + \cos B°)/2\right]^2/82.3 = \text{OHMS}$$

Note:

Antenna must be shorter than 90° electrical length.

Example:

Given A = 24 inches

B = 55 inches

C = 55 inches

a = .125 inches

F = 7.2 MHz

Fing L and R_{RAD}

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 010	
[XEQ] [ALPHA] LVA [ALPHA]	A?
24 [R/S]	B?
55 [R/S]	C?
55 [R/S]	a?
.125 [R/S]	F?
7.2 [R/S]	L=41.24
[R/S]	R=3.87

User Instructions

SIZE: 010

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load Program			
2	Begin execution		[XEQ] LVA	A?
	--input A (inches)	A	[R/S]	B?
	--input B (inches)	B	[R/S]	C?
	--input C (inches)	C	[R/S]	a?
	--input a (inches)	a	[R/S]	F?
	--input F (MHz)	F	[R/S]	L=
			[R/S]	R=

Program Listings

```
01♦LBL "LVA"        Initialize
02 "A?"
03 PROMPT
04 STO 01
05 "B?"
06 PROMPT
07 STO 02
08 "C?"
09 PROMPT
10 STO 03
11 "a?"
12 PROMPT
13 STO 04
14 "F?"
15 PROMPT
16 STO 05
17 RCL 01
18 RCL 02
19 RCL 03
20 +
21 +
22 RCL 04
23 /
24 LOG
25 138              Zo
26 *
27 RCL 03
28 RCL 05
29 32.8
30 /
31 STO 08
32 *
33 STO 06           C° → R06
34 TAN
35 1/X
36 RCL 02
37 RCL 08
38 *
39 STO 07           B° → R07
40 TAN
41 -
42 *
43 2
44 /
45 PI
46 /
47 RCL 05
48 /
49 "L="             L in µ Hy
50 XEQ d
51 RCL 07
52 COS
53 ENTER↑
54 ENTER↑
55 RCL 06
56 *
57 2
58 /
59 X<>Y
60 1
61 +
62 RCL 07
63 *
64 2
65 /
66 +
67 ENTER↑
68 *
69 82.3
70 /
71 "R="             R in Ohms
72♦LBL d
73 FIX 2            Display
74 ARCL X
75 AVIEW
76 STOP
77 RTN
78 .END.
80
90
00
```

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00		50	
	A		
	B		
	C		
	a		
05	F	55	
	C°		
	B°		
	Temp storage		
	Temp storage		
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 010 TOT. REG. 26 USER MODE
ENG ____ FIX 2 SCI ____ ON ____ OFF ____
DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
		none	

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY
none			

LOADED DIPOLE ANTENNAS

The required loading inductance is:

$$L(\mu Hy) = \frac{10^6}{68\pi^2 f^2} \quad \frac{[\ln(wy) - 1]\left[\frac{y^2}{z^2} - 1\right]}{y} - \frac{\left[\frac{x^2}{z^2} - 1\right][\ln(xw) - 1]}{x}$$

where $z = 234/f$

$x = A/2 - B$

$y = z - B$

$w = 24/DIA$

and A = See sketch

B = See sketch

f = Frequency in MHz

DIA = Diameter in inches

Note: $A < 180°$ Electrical length.

Sketch:



Example:

Given an antenna with the following dimensions:

A = 130 feet   Frequency = 1.8 MHz

B = 16.5 feet   DIA = 0.1 inch

Find the required coil inductance for resonance.

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 010	
[XEQ] [ALPHA] LDA [ALPHA]	A?
130 [R/S]	B?
16.5 [R/S]	F?
1.8 [R/S]	d?
0.1 [R/S]	L=59.38

User Instructions

SIZE: 010

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load program			
2	Begin execution		[XEQ] LDA	A?
	--input A (feet)	A	[R/S]	B?
	--input B (feet)	B	[R/S]	F?
	--input F (MHz)	F	[R/S]	d?
	--input DIA	DIA	[R/S]	L=

Program Listings

Program	Comment
01◆LBL "LDA"	Initialize
02 "A?"	
03 PROMPT	
04 2	
05 /	
06 STO 07	
07 "B?"	
08 PROMPT	
09 STO 01	
10 "F?"	
11 PROMPT	
12 STO 06	
13 234	
14 RCL 06	
15 /	
16 STO 03	$z \rightarrow R_{03}$
17 "d?"	
18 PROMPT	
19 STO 09	DIA $\rightarrow R_{09}$
20 RCL 07	
21 RCL 01	
22 -	
23 STO 02	$x \rightarrow R_{02}$
24 RCL 03	
25 RCL 01	
26 -	
27 STO 04	
28 RCL 03	
29 /	
30 ENTER↑	
31 *	
32 1	
33 -	
34 STO 08	$\frac{y^2}{z^2} - 1 \rightarrow R_{08}$
35 RCL 09	
36 24	
37 X<>Y	
38 /	
39 STO 05	
40 RCL 04	$w \rightarrow R_{05}$
41 *	
42 LN	
43 1	
44 -	$\ln (wy) - 1$
45 RCL 08	
46 *	
47 RCL 04	
48 /	
49 RCL 02	
50 RCL 03	

Program	Comment
51 /	
52 ENTER↑	
53 *	
54 1	
55 -	$(\frac{x}{z})^2 - 1$
56 RCL 02	
57 RCL 05	
58 *	
59 LN	
60 1	
61 -	$\ln (xw) - 1$
62 *	
63 RCL 02	
64 /	
65 -	
66 1 E6	
67 *	
68 68	
69 /	
70 RCL 06	
71 PI	
72 *	
73 ENTER↑	$\pi^2 f^2$
74 *	
75 /	
76 FIX 2	
77 "L="	Display L
78 ARCL X	
79 AVIEW	
80 .END.	

90

00

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00		50	
	B		
	x		
	z		
	y		
05	w	55	
	FREQ		
	A/2		
	Temp		
	DIA		
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 010 TOT. REG. 25 USER MODE
ENG ___ FIX 2 SCI ___ ON ___ OFF X
DEG ___ RAD ___ GRAD ___

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
		None	

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY
None			

GAIN OF A HORIZONTAL RHOMBIC ANTENNA AT ZERO AZIMUTH

This program estimates the on-axis gain of a horizontal rhombic antenna, placed above real earth, for specified take-off angles (TOA).

Inputs are: The antenna's height in meters (H); the leg length in meters (L); the tilt angle in degrees (ϕ); the frequency of operation in MHz (F); the lowest (initial) TOA, the step (increment) of TOA, and the highest (largest) TOA to be used, all in integer degrees; the earth's conductivity in mho/meter (σ); and the earth's relative dielectric constant (ε_R).

The program uses the formulation of ESSA Technical Report ERL110-ITS78.

Note: The program contains no testing for invalid inputs.
Only integer values of the take-off angle are used.

Example:

Antenna characterized by: H = 20m L = 114m $\phi = 70°$

Earth characterized by: σ = 0.001 mho/m, $\varepsilon_R = 4$

Wish to evaluate starting at TOA = 6°, in increments of 6° until TOA = 24° at a frequency of 16 MHz.

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 024	
[XEQ] [ALPHA] RHOMBIC [ALPHA]	H?
20 [R/S]	L?
114 [R/S]	∡?
70 [R/S]	FREQ?
16 [R/S]	TOA L?
6 [R/S]	TOA S?
6 [R/S]	TOA H?
24 [R/S]	S?
.001 [R/S]	E?
4 [R/S]	∡=6.0 G=18.7
[R/S]	∡=12.0 G=20.8
[R/S]	∡=18.0 G=15.0
[R/S]	∡=24.0 G=-7.2

User Instructions

SIZE: 024

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load Program			
2	Begin execution, Load data		[XEQ] Rhombic	H?
	--Height	H(Meters)	[R/S]	L?
	--Leg Length	L(Meters)	[R/S]	∡?
	--Tilt angle	ϕ(Degrees)	[R/S]	FREQ?
	--Frequency	F (MHz)	[R/S]	TOA L?
	--Lowest (initial) Take-off angle	(Degrees)	[R/S]	TOA S?
	--Take-off angle step	(Degrees)	[R/S]	TOA H?
	--Highest Take-off angle	(Degrees)	[R/S]	S?
	--Earth's Conductivity	σ mho/meter	[R/S]	E?
	--Earth's dielectric constant	ε_R		∡=
			[R/S]	G =
				∡=
			[R/S]	G =
				:
				(until
				complete)

Program Listings

01♦LBL "RHOMBIC"	Initialize	51 -	
02 "H?"		52 X>0?	
03 PROMPT		53 RTN	
04 STO 22		54 RCL 05	
05 "L?"		55 RCL 04	
06 PROMPT		56 GTO 02	
07 STO 23		57♦LBL 00	Gain subroutine
08 "∡?"		58 COS	Δ in x
09 PROMPT		59 STO 11	F in y
10 D-R		60 LASTX	
11 STO 00		61 SIN	
12 "FREQ?"		62 STO 12	
13 PROMPT		63 RAD	
14 STO 05		64 RDN	
15 "TOA L?"		65 RDN	
16 PROMPT		66 299.8	
17 STO 01		67 X<>Y	
18 "TOA S?"		68 /	
19 PROMPT		69 STO 13	
20 STO 02		70 PI	
21 "TOA H?"		71 X<>Y	
22 PROMPT		72 /	
23 STO 03		73 STO 14	
24 "S?"		74 1	
25 PROMPT		75 RCL 11	
26 STO 20		76 RCL 00	
27 "E?"		77 SIN	
28 PROMPT		78 *	
29 STO 21		79 -	
30 RCL 05		80 STO 15	
31 RCL 01		81 RCL 20	Start to calcu-
32 STO 04		82 RCL 13	late reflection
33♦LBL 02	Go calculate	83 *	coefficient
34 XEQ 00	gain	84 -60	
35 RCL 04		85 *	
36 FIX 1	Display angle	86 RCL 21	
37 "∡="		87 RCL 11	
38 ARCL X		88 X↑2	
39 AVIEW		89 -	
40 PSE		90 R-P	
41 X<>Y		91 SQRT	
42 "G="	Display G	92 X<>Y	
43 ARCL X		93 2	
44 AVIEW		94 /	
45 STOP		95 X<>Y	
46 RCL 02		96 P-R	
47 RCL 04	Increment D	97 STO 16	
48 +		98 RDN	
49 STO 04		99 STO 17	
50 RCL 03		100 CHS	
		101 R↑	

Program Listings

```
102 CHS
103 RCL 12
104 +
105 R-P
106 X<>Y
107 RCL 16
108 RCL 12
109 +
110 RCL 17
111 X<>Y
112 R-P
113 RDN
114 -
115 STO 17
116 X<>Y
117 R↑
118 /
119 STO 16
120 RCL 17
121 RCL 12
122 RCL 14
123 *
124 RCL 22
125 *
126 4
127 *
128 -
129 COS
130 *
131 2
132 *
133 1
134 +
135 RCL 16
136 X↑2
137 +
138 RCL 15
139 RCL 14
140 *
141 RCL 23
142 *
143 SIN
144 X↑2
145 RCL 00
146 COS
147 *
148 RCL 15
149 /
150 X↑2
151 *
152 2.16
153 *
154 LOG
155 10
156 *
157 DEG
158 RTN
159 .END.
```

Gain is in X

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	ϕ	50	
	TOA L		
	TOA S		
	TOA H		
	TOA in use		
05	F	55	
10		60	
	COS Δ		
	SIN Δ		
	λ		
	π/λ		
15	U	65	
	$\lvert R_H \rvert$ $\lvert R_H \rvert$		
	$\measuredangle R_H$		
20	σ	70	
	ϵ_R		
	H		
	L		
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 024 TOT. REG. 58 USER MODE
ENG ___ FIX 1 SCI ___ ON ___ OFF X
DEG X RAD ___ GRAD ___

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
		None	

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY
None			

AZIMUTH PATTERN OF CYLINDRICAL ARRAY OF ANTENNAS

This program calculates the angle at which the individual radiating element is to be evaluated and with this information the user inputs the magnitude and phase of the individual radiating element at this angle. The program then calculates the phase of the individual contribution to the total radiated field, with reference to the center of rotation of the structure (point 0 in sketch). The program then repeats these steps for all of the individual elements, and sums their individual contributions to determine the resultant field at a given observation angle, θ. The variables are:

α_n = pattern evaluation angle
R = radius of antenna array;
θ_f = angle between the face normals;
γ = skew angle in C.W. direction of the individual radiating element's reference azimuth from face normal
N = total number of individual radiating elements
n = number of individual elements
δ = offset distance of individual element from center of face
ϕ_s = phase of contribution of an individual element due to its location in space
$M(\alpha)$, $p(\alpha)$ = magnitude and phase of individual element evaluated at α
Ψ_n = phase of signal to face n
A_n = magnitude of signal to face n

The equations are:

$$M(\theta) = \text{abs}\left[\sum_{n=1}^{N} An \cdot M(\alpha_n) \exp\left(j\Phi(\alpha_n)\right)\right]$$

where

$$\Phi = \phi_s + p + \Psi$$

$$\alpha_n = \theta_n - \gamma$$

$$\phi_s = 360/\lambda(R\cos\theta_n + \delta\sin\theta_n)$$

$$\theta_n = \theta - \theta_f(n-1)$$

Note: $M(\theta)$ must be renormalized after computation.





Pattern of Individual Element



Detail of the individual elements offset and skew.

Example:

What is the radiated field at an azimuth of 10°? The array is a four sided structure with a side of 6 m. The antennas are offset 2 m from the center of the panel towards the right when viewed from the center of the array. Each element is skewed 10° C.W. when viewed from the top. $\lambda = 2.2$ m. The elements, in order from n=1 to n=4, are fed as follows:

$$1.0\angle 0°,\ 0.9\angle 90°,\ 0.9\angle 180°,\ 1.0\angle 270°.$$

The pattern of the individual element is as follows (azimuth, magnitude, phase in that order):

0°, $1.0\angle 0°$; 10°, $0.94\angle 0°$; 40°, $0.54\angle 5°$; 90°, $0.05\angle 10°$;

130°, $0.02\angle 30°$; 180°, $0.08\angle -50°$; 220°, $0.02\angle 30°$; 270°,

$0.05\angle 7°$; 310°, $0.41\angle 6°$.

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 025	
[XEQ] [ALPHA] AZIPAT [ALPHA]	N?
4 [R/S]	∡F?
90 [R/S]	R?
3 [R/S]	SCEW A.?
10 [R/S]	SCEW L.?
2 [R/S]	WAV.L?
2.2 [R/S]	∡?
10 [R/S]	FACE 1
	∡N=0.00
[R/S]	PHASE R?
0 [R/S]	PHASE N?
0 [R/S]	AMP. R?
1 [R/S]	AMP. N?
1 [R/S]	FACE 2
	∡N= 270.00
[R/S]	PHASE R?
7 [R/S]	PHASE N?
90 [R/S]	AMP. R?
.05 [R/S]	AMP. N?
.9 [R/S]	FACE 3
	∡N=180.00
[R/S]	PHASE R?
50 [CHS] [R/S]	PHASE N?
180 [R/S]	AMP. R?
.08 [R/S]	AMP. N?
.9 [R/S]	FACE 4
	∡N=90.00
[R/S]	PHASE R?
10 [R/S]	PHASE N?
270 [R/S]	AMP. R?
.05 [R/S]	AMP. N?
1 [R/S]	M=3.22
[R/S]	∡=-162.56

User Instructions

SIZE: 025

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1.	Load Program			
2.	Begin execution, input:		[XEQ]AZIPAT	N?
	--Number of elements in array	N	[R/S]	$\measuredangle$ F?
	--Angle between faces	θ_f, deg.	[R/S]	R?
	--radius of array	R	[R/S]	SCEW A.?
	--angular scew	γ, deg.	[R/S]	SCEW L.?
	--lateral scew	δ	[R/S]	WAV.L.?
	--wave length	λ	[R/S]	$\measuredangle$?
3.	Input desired azimuth angle (calculator	θ, deg.	[R/S]	FACE (n)
	will pause with face number and then			$\measuredangle$N=
	display the individual pattern evaluation			
	angle, α_n)			
4.	--continue, input:		[R/S]	PHASE R?
	--phase of radiated field of α_n	P (α_n)	[R/S]	PHASE N?
	--phase of signal to face N	Ψ_N	[R/S]	AMP. R?
	--amplitude of radiated field	$m(\alpha_n)$	[R/S]	AMP. N?
	--amplitude of signal to face N	A_N	[R/S]	FACE (n)
				$\measuredangle$N=
	Go to Step 4			
	If last n	A_N	[R/S]	M=
			[R/S]	$\measuredangle$=

Program Listings

Program	Comment
01◆LBL "AZI	Initialize
PAT"	
02 "N?"	
03 PROMPT	
04 "∡F?"	
05 PROMPT	
06 "R?"	
07 PROMPT	
08 STO 20	
09 RDN	
10 STO 21	
11 RDN	
12 STO 22	
13 XEQ 01	
14 "SCEW A.	
?"	
15 PROMPT	
16 "SCEW L.	
?"	
17 PROMPT	
18 STO 23	
19 RDN	
20 STO 24	
21 XEQ 01	
22 "WAV. L?	
"	
23 PROMPT	
24 STO 09	
25 XEQ 01	Calculate frequency sensitive constants
26 GTO D	
27◆LBL 01	
28 CLX	
29 360	
30 ENTER↑	
31 ENTER↑	
32 RCL 20	
33 *	
34 RCL 09	
35 X=0?	
36 RTN	
37 /	
38 STO 08	
39 X<>Y	
40 RCL 23	
41 *	
42 RCL 09	
43 /	
44 STO 07	
45 CLX	
46 RTN	Pattern Calculations
47◆LBL D	
48 "∡?"	
49 PROMPT	Initialize
50 STO 05	
51 1	
52 STO 06	
53 CLX	
54 STO 04	
55 STO 03	
56◆LBL 10	Calculate Pattern
57 RCL 05	
58 COS	
59 RCL 08	
60 *	
61 RCL 05	
62 SIN	
63 RCL 07	
64 *	
65 +	
66 RCL 05	
67 RCL 24	
68 -	
69 X<0?	
70 XEQ 03	
71 RCL 06	
72 FIX 0	
73 "FACE "	Display FACE
74 ARCL X	
75 AVIEW	
76 PSE	
77 FIX 2	
78 RDN	
79 "∡N="	Display N
80 ARCL X	
81 AVIEW	
82 STOP	
83 "PHASE R	input: phase of radiated field phase of signal to face N
?"	
84 PROMPT	
85 "PHASE N	
?"	
86 PROMPT	
87 +	
88 R↑	
89 +	
90 "AMP. R?	input: amplitude of radiated field amplitude of signal to face N
"	
91 PROMPT	
92 "AMP. N?	
"	
93 PROMPT	
94 +	

Program Listings

Program	Comments
95 P-R	
96 ST+ 03	
97 X<>Y	
98 ST+ 04	
99 RCL 22	
100 RCL 06	
101 X=Y?	
102 GTO 02	
103 1	
104 ST+ 06	
105 RCL 21	
106 ST- 05	
107 GTO 10	
108♦LBL 02	
109 RCL 04	
110 RCL 03	
111 R-P	
112 "M="	Display M
113 ARCL X	
114 AVIEW	
115 STOP	
116 X<>Y	
117 "∡="	Display
118 ARCL X	
119 AVIEW	
120 STOP	
121 GTO D	
122♦LBL 03	Change all angles to positive
123 360	
124 +	
125 X<0?	
126 GTO 03	
127 RTN	
128 .END.	

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

		50	
	Re E Field		
	Im E Field		
05	θ_n	55	
	n		
	360 δ/λ		
	360 R/λ		
	λ		
10		60	
15		65	
20	R	70	
	θ_f		
	N		
	δ		
	γ		
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 025 TOT. REG. 61 USER MODE
ENG ___ FIX 2 SCI ___ ON ___ OFF X
DEG X RAD ___ GRAD ___

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
		None	

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY
None			

COLINEAR ANTENNA GAIN AND PATTERN

This program performs numerical integration of:

$$G = \int_0^{90} f(\theta)^2 \sin\theta \, d\theta$$

with steps of θ_i.

$$f(\theta)^2 = f(\theta)^2_{dipole} \; f(\theta)^2_{array}$$

$$f(\theta)_{dipole} = \frac{\cos(\pi L_\lambda \cos\theta) - \cos(\pi L_\lambda)}{\sin(\theta)(1-\cos(\pi L_\lambda) + 1 \times 10^{-9})}$$

(the term $(1-\cos(\pi L_\lambda))$ normalizes $f(\theta)_{dipole}$.)

$$f(\theta)_{array} = \frac{\sin(N\Psi/2)}{N \sin(\Psi/2)} \qquad \Psi = 2\pi(S_\lambda + L_\lambda)\cos\theta + \Psi_o$$

where

Ψ = Progressive element-to-element phase shift in radians. For broad-side array, $\Psi_o = 0$. For $\Psi_o = 0$ use both positive and negative value to evaluate both sides of pattern.

L_λ = Total dipole length in wavelengths.

S_λ = Tip-to-tip dipole spacing in wavelengths.

N = Total number of elements in array.

Note: Valid only for equal amplitude feed, and uniform spacing and dipole length. Valid for any length center fed dipole, or end fed dipole of one-half wavelength or less.

Example:

1. Calculate the gain for a ½λ dipole. (N=3)

2. Calculate the gain of two 5/4λ dipoles spaced ½λ tip-to-tip. (N=2)

	Keystrokes:	Display:
	[XEQ] [ALPHA] SIZE [ALPHA] 011	
1.	[///] [CF] 01	
	[///] [SF] 02	
	[XEQ] [ALPHA] CAGP [ALPHA]	PHASE SHIFT?
	0 [R/S]	DIPOLE L.?
	.5 [R/S]	DIPOLE S.?
	0 [R/S]	N?
	3 [R/S]	G=2.15
2.	[///] [CF] 01	
	[///] [CF] 02	
	[XEQ] [ALPHA] CAGP [ALPHA]	PHASE SHIFT?
	0 [R/S]	DIPOLE L.?
	1.25 [R/S]	DIPOLE S.?
	.5 [R/S]	N?
	2 [R/S]	G=8.33

User Instructions

SIZE: 011

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load Program			
2	To speed dipole only or array factor			
	only computations			
	--dipole only		CF01	
			SF02	
	--array factor only		SF01	
			CF02	
	otherwise		CF01	
			CF02	
3	Begin execution and input		[XEQ]CAGP	PHASE SHIFT?
	--Progressive phase shift (radians)	Ψ_o	[R/S]	DIPOLE L.?
	--Dipole length (wavelengths)	L_λ	[R/S]	DIPOLE S.?
	--Dipole spacing (tip-to-tip) (wavelengths)	S_λ	[R/S]	N?
	--Number of dipoles in array	N	[R/S]	G=

Program Listings

Initialization

```
01◆LBL "CAG
P"
02 RAD
03 SF 00
04 2
05 XEQ 07
06 STO 10
07 "PHASE S
HIFT?"
08 PROMPT
09 STO 00
10 "DIPOLE
L.?"
11 PROMPT
12 STO 01
13 "DIPOLE
S.?"
14 PROMPT
15 STO 02
16 "N?"
17 PROMPT
18 STO 03
19 XEQ B
20 "∡?"
21 PROMPT
```

Calculates number of steps for °/step

```
22 CF 00
23◆LBL C
24 D-R
25 STO 08
26 0
27 STO 06
28 STO 07
29 PI
30 2
31 /
32 RCL 08
33 /
34 FIX 0
35 RND
36 STO 05
37 FIX 2
38 FS? 00
39 RTN
40 PSE
41 GTO D
```

advance starting angle

```
42◆LBL c
43 D-R
44 RCL 08
45 /
46 FIX 0
47 RND
48 STO 06
49 RTN
```

Calculates minimum number of steps

```
50◆LBL B
51 SF 00
52 90
53 RCL 01
54 RCL 03
55 *
56 RCL 03
57 1
58 -
59 RCL 02
60 *
61 +
62 PI
63 2
64 /
65 *
66 3
67 +
68 /
69 XEQ C
```

Calculates gain

```
70◆LBL b
71 XEQ 09
72 XEQ 01
73 RCL 09
74 SIN
75 *
76 ST+ 07
77 1
78 ST+ 06
79 RCL 05
80 RCL 06
81 X≠Y?
82 GTO b
83 RCL 07
84 .5
85 -
86 RCL 08
87 *
88 XEQ 07
89 CHS
90 STO 04
```

Display G (final)

```
91 "G="
92 ARCL X
93 AVIEW
94 STOP
95 RTN
96◆LBL D
97 XEQ 08
98 R-D
```

Program Listings

Program	Comment
99 "∡="	Display
100 ARCL X	
101 AVIEW	
102 STOP	
103 XEQ 09	
104 XEQ 01	
105 XEQ 07	
106 "G="	Display G
107 ARCL X	
108 AVIEW	
109 STOP	
110 1	
111 ST+ 06	
112 RCL 05	
113 RCL 06	
114 X<=Y?	
115 GTO D	
116 GTO B	
117◆LBL 09	
118 XEQ 08	Brings θ° to
119 PI	Broadside
120 2	
121 /	
122 +	
123 STO 09	
124 RTN	
125◆LBL 08	
126 RCL 08	Calculate angle
127 RCL 06	θ
128 *	
129 RTN	
130◆LBL 07	
131 E-9	
132 +	
133 LOG	
134 10	
135 *	
136 RTN	
137◆LBL 01	$f(\theta)_b^2 \cdot f(\theta)_k^2$
138 FS? 01	
139 GTO 02	
140 RCL 09	
141 COS	
142 PI	
143 *	
144 RCL 01	
145 *	
146 COS	
147 RCL 01	
148 PI	
149 *	

```
150 COS
151 -
152 1
153 LASTX
154 -
155  E-9
156 +
157 /
158 RCL 09
159 SIN
160 /
161 X↑2
162 FS? 02
163 RTN
164◆LBL 02
165 RCL 09
166 COS
167 RCL 02
168 RCL 01
169 +
170 *
171 PI
172 *
173 2
174 *
175 RCL 00
176 +
177 2
178 /
179 STO 04
180 RCL 03
181 *
182 SIN
183 RCL 04
184 SIN
185 /
186 RCL 03
187 /
188 X↑2
189 FC? 01
190 *
191 RTN
192 .END.
```

00	

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	Ψ_o	50	
	L_λ		
	S_λ		
	N		
	Temp storage		
05	n	55	
	1. 2, 3, ...n		
	$\Sigma\ f(\theta)^2 \sin\theta$		
	θ		
10	dB/step	60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 011 TOT. REG. 54 USER MODE
ENG ___ FIX 2 SCI ___ ON ___ OFF X
DEG ___ RAD ___ GRAD ___

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
00		Return	Continue
01		Array only	
02		Dipole only	

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY
None			

BEAM PATTERN FOR UNIFORM ARRAY

This program computes the normalized beam pattern of a uniformly spaced and weighted discrete linear array for an arbitrary number of sensors, arbitrary steering angle, and arbitrary wavelength.

The normalized beam pattern is given by:

$$R(\theta_i) = 10 \log_{10} \left(\frac{\sin \left[\frac{N180\ d}{\lambda} (\sin \theta_i - \sin \theta_s) \right]}{N \sin \left[\frac{180\ d}{\lambda} (\sin \theta_i - \sin \theta_s) \right]} \right)^2$$

where
- N = total number of sensors
- d = inter sensor spacing
- λ = wavelength of incident plane wavefront
- θ_i = angle of incident wavefront
- θ_s = steering angle of array
- R = response in decibels



Note: All angles are assumed to be measured in degrees. The quantities d and λ must have the same units.

Reference: "Principles of Aperture and Array System Design", Steinberg John Wiley and Sons, 1976.

Example:

Given a five sensor array with an inter sensor spacing of 50 feet, steered to broadside ($\theta_s = 0$), calculate the normalized response for a plane wave of 100 feet incident at 0°, 5°, 10°, 15°, and 20°.

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 009	
[XEQ] [ALPHA] BMPAT [ALPHA]	N?
5 [R/S]	d?
50 [R/S]	L?
100 [R/S]	∡S?
0 [R/S]	∡I?
0 [R/S]	R=0.00 dB
[R/S]	∡I?
5 [R/S]	R=-0.66 dB
[R/S]	∡I?
10 [R/S]	R=-2.77 dB
[R/S]	∡I?
15 [R/S]	R=-6.88 dB
[R/S]	∡I?
20 [R/S]	R=-15.30 dB

User Instructions

SIZE: 009

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load Program			
2	Begin execution and input		[XEQ] BMPAT	N?
	--total sensors	N	[R/S]	d?
	--inter sensor spacing	d	[R/S]	L?
	--wavelength	λ	[R/S]	∡S?
	--steering angle	θ_s (deg)	[R/S]	∡I?
3	Input angle of incident wavefront	θ_i (deg)	[R/S]	R= dB
4	To use a new angle of incident go to		[R/S]	∡I?
	Step 3			

Program Listings

01♦LBL "BMPAT"	Initialize
02 FIX 2	
03 "N?"	
04 PROMPT	
05 STO 04	
06 "d?"	
07 PROMPT	
08 STO 05	
09 "L?"	
10 PROMPT	
11 STO 06	
12 "∡S?"	
13 PROMPT	
14 SIN	
15 STO 08	
16♦LBL A	
17 "∡I?"	
18 PROMPT	
19 SIN	
20 STO 07	Computes beam
21♦LBL C	former sum for
22 180	one value of
23 RCL 05	θ_i
24 *	
25 RCL 06	
26 /	
27 STO 00	
28 RCL 07	
29 RCL 08	
30 -	
31 STO 01	
32 *	
33 RCL 04	
34 *	
35 SIN	
36 RCL 01	
37 RCL 00	
38 *	
39 SIN	
40 RCL 04	
41 *	
42 X=0?	Test for 0
43 GTO 00	
44 /	
45 ABS	
46 LOG	
47 20	
48 *	
49♦LBL B	
50 "R="	Display R
51 ARCL X	
52 "⊦ dB"	
53 AVIEW	
54 STOP	
55 GTO A	
56♦LBL 00	
57 0	
58 GTO B	
59 RTN	
60 .END.	

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	Temp storage	50	
	Temp storage		
	N		
05	d	55	
	λ		
	$\sin\theta_i$		
	$\sin\theta_s$		
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 009 TOT. REG. 24 USER MODE

ENG __ FIX 2 SCI __ ON __ OFF X

DEG X RAD __ GRAD __

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
		None	

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY
None			

RADAR ANTENNA BEAMWIDTH AND GAIN

Rectangular Antenna

$$\text{Beamwidth}_x = \frac{k\lambda}{L_x} \qquad \lambda = \frac{299.8}{f}$$

$$\text{Beamwidth}_y = \frac{k\lambda}{L_y} \qquad G = 10 \text{ LOG } \frac{27000}{BW_x \cdot BW_y}$$

Circular Antenna

$$\text{Beamwidth} = \frac{k\lambda}{d} \qquad G = 10 \text{ LOG } \frac{27000}{BW}$$

L_x, L_y, d in meters, f in MHz

Weighting	Rectangular	Circular
Uniform	k= 51	k= 58
Cosin	k= 68	k= 76
Hamming	k= 74.5	k= 80
Cos on 10 dB Ped.		
$1-R^2$		k= 72.5
$(1-R^2)^2$		k= 84.3

On calculated gain:

1. Subtract 1.5 dB for $\csc^2$ beams.
2. Add 1.5 dB for 2 dimensional arrays.
3. Add 0.5 dB for linear array fed parabolic cylinders.

Example:

1. Find antenna beamwidths and gain of a uniformly weighted array operating at 500 MHz. $L_x = 50$ m, $L_y = 25$ m.
2. Find antenna beamwidth and gain of a cosin weighted 10 meter circular antenna at 3000 MHz.

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 004	
[XEQ] [ALPHA] REC [ALPHA]	FREQ.?
500 [R/S]	X?
50 [R/S]	Y?
25 [R/S]	
[XEQ] [ALPHA] U [ALPHA]	BWX=0.61
[R/S]	BWY=1.22
[R/S]	G=45.57
1.5 [+]	47.07 (actual gain)
[XEQ] [ALPHA] CIR [ALPHA]	FREQ.?
3000 [R/S]	d?
10 [R/S]	
[XEQ] [ALPHA] C [ALPHA]	BW=0.76
[R/S]	G=46.70

User Instructions

SIZE: 004

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load Program			
2	Begin execution:			
	For rectangular antenna		[XEQ] REC	FREQ.?
	--input frequency (MHz)	f	[R/S]	x?
	--input dimension x	x	[R/S]	y?
	--input dimension y	y	[R/S]	
	Go to 3			
	For circular antenna		[XEQ] CIR	FREQ?
	--input frequency (MHz)	f	[R/S]	d?
	--input diameter	d	[R/S]	
3	Weighting			
	--for uniform		[XEQ] U	
	--for cosin		[XEQ] C	
	--for hamming		[XEQ] H	
4	Beam width and gian are displayed			
	--rectangular			BWX=
			[R/S]	BWY=
			[R/S]	G=
	--Circular			BW=
			[R/S]	G=

Program Listings

01	♦LBL "REC"	Rectangular
02	CF 01	
03	GTO A	
04	♦LBL "CIR"	Circular
05	SF 01	
06	♦LBL A	
07	FIX 2	
08	"FREQ.?"	Initialization
09	PROMPT	
10	299.8	
11	X<>Y	
12	/	
13	STO 01	
14	"X?"	
15	FS? 01	
16	"d?"	
17	PROMPT	
18	FS? 01	
19	STOP	
20	"Y?"	
21	PROMPT	
22	STOP	
23	♦LBL "U"	uniform wt.
24	FS? 01	
25	GTO 01	
26	51	
27	GTO E	
28	♦LBL 01	
29	58.4	
30	GTO E	
31	♦LBL C	cosine wt.
32	FS? 01	
33	GTO 02	
34	68	
35	GTO E	
36	♦LBL 02	
37	76	
38	GTO E	
39	♦LBL H	haming wt
40	FS? 01	
41	GTO 03	
42	74.5	
43	GTO E	
44	♦LBL 03	
45	80	
46	♦LBL E	kλ
47	RCL 01	
48	*	
49	STO 02	
50	X<>Y	BWY or BW
51	/	
52	STO 03	
53	FS? 01	
54	GTO 09	
55	RDN	
56	RCL 02	
57	X<>Y	
58	/	
59	RCL 03	display
60	X<>Y	
61	"BWX="	BWX
62	ARCL X	
63	AVIEW	
64	STOP	
65	X<>Y	
66	♦LBL 08	
67	"BWY="	BWY or BW
68	FS? 01	
69	"BW="	
70	ARCL X	
71	AVIEW	
72	STOP	
73	*	
74	27 E3	
75	X<>Y	
76	/	
77	LOG	
78	10	
79	*	
80	"G="	Gain
81	ARCL X	
82	AVIEW	
83	STOP	
84	♦LBL 09	for circular
85	ENTER↑	
86	GTO 08	
87	.END.	

90	
00	

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	λ meters	50	
	k λ		
	BWY or BW		
05		55	
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 004 TOT. REG. 29 USER MODE

ENG ____ FIX 2 SCI ____ ON ____ OFF X

DEG X RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
01		Circular antenna	Rectangular ant.

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY
None			

ANTENNAS

Given the height, h, and width, ℓ, of an antenna, this program will compute the -3dB vertical and horizontal beamwidths, β_v and β_h respectively; equations for these quantities are:

$$\beta_v \simeq k(\frac{\lambda}{h})$$

$$\beta_h \simeq k(\frac{\lambda}{\ell})$$

where k = antenna taper factor (k=1.4 is used in this program)

λ = wavelength

Given the gain of antenna, G, this program will also compute the equivalent antenna area:

$$A_e = \frac{G\lambda^2}{4\pi}$$

The program will also compute the inverse of these quantities.

Note: λ should be small compared to antenna dimensions.



Example:

1. A plate antenna array operating at 6200 MHz (λ=0.16 ft) is 8.5 ft long and 1.9 ft high, assuming a taper factor of 1.4, what are the associated horizontal and vertical beamwidths?

2. If vertical and horizontal beamwidths of 6.5 and 2.5 degrees were desired, what antenna dimensions would be required for antenna in (1)?

3. A dish antenna operating at 8500 MHz (λ=.12 ft) has a measured gain of 28 dB. What is its equivalent area?

4. An antenna operating at 6500 MHz (λ=0.15 ft) has an area of 16.5 sq. ft. What is its expected gain?

	Keystrokes:	Display:
	[XEQ] [ALPHA] SIZE [ALPHA] 009	
1.	[XEQ] [ALPHA] BEAM [ALPHA]	L?
	8.5 [R/S]	H?
	1.9 [R/S]	WAV. L?
	0.16 [R/S]	BWH=1.51
	[R/S]	BWV=6.75
2.	[XEQ] [ALPHA] H-L [ALPHA]	BWV?
	6.5 [R/S]	BWH?
	2.5 [R/S]	WAV. L?
	.16 [R/S]	H=1.97
	[R/S]	L=5.13
3.	[XEQ] [ALPHA] AREA [ALPHA]	G?
	28 [R/S]	WAV. L?
	.12 [R/S]	AREA=0.72
4.	[XEQ] [ALPHA] GAIN [ALPHA]	Ae?
	16.5 [R/S]	WAV. L?
	.15 [R/S]	G=39.65

User Instructions

SIZE: 009

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load Program			
2	For beamwidth:		[XEQ] BEAM	L?
	--input antenna length	L	[R/S]	H?
	--input antenna height	H	[R/S]	WAV. L?
	--input wave length	λ	[R/S]	BWH=
	--beam width is displayed		[R/S]	BWV=
3	For antenna height and length		[XEQ] H-L	BWV?
	--input vertical beamwidth	β_v	[R/S]	BWH?
	--input horizontal beamwidth	β_h	[R/S]	WAV. L?
	--input wave length	λ	[R/S]	H=
	--height and length are displayed		[R/S]	L=
4	For equivalent area		[XEQ] AREA	G?
	--input gain (dB)	G	[R/S]	WAV. L?
	--input wave length	λ	[R/S]	AREA=
5	For gain		[XEQ] GAIN	Ae?
	--input Area	A_e	[R/S]	WAV. L?
	--input wave length	λ	[R/S]	G=

Program Listings

```
01◆LBL "BEAM"          Beam
02 CF 01
03 GTO 00
04◆LBL "H-L"           H-L
05 SF 01
06◆LBL 00              Initialize
07 FIX 2
08 180
09 PI
10 /
11 1.4
12 *
13 STO 01
14 "L?"
15 FS? 01
16 "BWV?"
17 PROMPT
18 STO 03
19 "H?"
20 FS? 01
21 "BWH?"
22 PROMPT
23 STO 02
24 "WAV. L?"
25 PROMPT
26 STO 08
27 RCL 03
28 /
29 RCL 01
30 *
31 "BWH="              display BWH or H
32 FS? 01
33 "H="
34 ARCL X
35 AVIEW
36 STOP
37 RCL 01
38 RCL 02
39 /
40 RCL 08
41 *
42 "BWV="              display BWV or L
43 FS? 01
44 "L="
45 ARCL X
46 AVIEW
47 STOP
48◆LBL "AREA"          Area
49 CF 01
50 GTO 00
51◆LBL "GAIN"          Gain
52 SF 01
53◆LBL 00              Initialize
54 "G?"
55 FS? 01
56 "Ae?"
57 PROMPT
58 STO 06
59 "WAV. L?"
60 PROMPT
61 ENTER↑
62 *
63 RCL 06
64 FS? 01
65 GTO 00
66 10
67 /
68 10↑X
69 *
70 4
71 /
72 PI
73 /
74 "AREA="             Display AREA
75 ARCL X
76 AVIEW
77 STOP
78◆LBL 00
79 4
80 *
81 PI
82 *
83 /
84 1/X
85 LOG
86 10
87 *
88 "G="                display G
89 ARCL X
90 AVIEW
91 STOP
92 .END.
```

00

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	180k/π	50	
	h		
	ℓ		
	β_v		
	β_h		
05	Temp. storage	55	
	Temp. storage		
	λ		
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 009 TOT. REG. 36 USER MODE
ENG ___ FIX 2 SCI ___ ON ___ OFF X
DEG X RAD ___ GRAD ___

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
01		H-L or GAIN	BEAM or AREA

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

PARABOLIC ANTENNA CALCULATIONS

$$G = f20\log(D) + 10\log(10.2\,E) + 20\log(f)$$

$$D = \log^{-1}\left(\frac{G - 20\log(f) - 10\log(10.2\,E)}{20}\right)$$

$$\theta_{3dB} = \frac{66.4}{fD}$$

$$\theta_{NdB} = \left[\frac{\theta_{3dB}^{2}\downarrow NdB}{3}\right]^{1/2}$$

$$\downarrow NdB = \frac{12\alpha^2}{\theta_{3dB}^{2}}$$

where D = antenna diameter, feet

f = frequency, GHz

E = efficiency, decimal

α = off-axis angle, deg.

G = gain over isotropic, dB

θ_{3dB} = 3dB beamwidth, deg.

θ_{NdB} = NdB beamwidth, deg.

$\downarrow_{NdB}$ = off-axis dB down

Note: All calculations are based on the main beam only. Side lobes are not considered.

Example:

Perform the following calculations for a 10 foot parabolic antenna at 6.175 GHz:

1. Gain (55% efficiency)
2. 3dB beamwidth
3. 15dB beamwidth
4. dB down at 1 degree off-axis

Keystrokes:	Display:	
[XEQ] [ALPHA] SIZE [ALPHA] 007		
[///] [SF] 00		
[///] [CF] 01		
[XEQ] [ALPHA] GAIN [ALPHA]	FREQ.?	
6.175 [R/S]	% EFF.?	
55 [R/S]	DIA. ?	
10 [R/S]	dB1=43.30	(1)
[XEQ] [ALPHA] 3dB [ALPHA]	DIA. ?	
10 [R/S]	∡ =1.08	(2)
[XEQ] [ALPHA] NdB [ALPHA]	N?	
15 [R/S]	∡ =2.40	(3)
[XEQ] [ALPHA] dB [ALPHA]	∡?	
1 [R/S]	dB=10.38	(4)

User Instructions

SIZE: 007

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load Program			
2	Initialize flags			
	--for new problem		SF 00	
	--for data in feet		CF 01	
	--for data in meters		SF 01	
3	To compute gain		[XEQ] GAIN	FREQ.?
	--input frequency (GHz)	f	[R/S]	% EFF.?
	--% EFFICIENCY	% E	[R/S]	DIA. ?
	--input antenna diameter	DIA	[R/S]	dB1=
4.	To compute antenna diameter		[XEQ] DIA	FREQ.?
	--input frequency (GHz)	f	[R/S]	% EFF.?
	--%EFFICIENCY	%E	[R/S]	dB1?
	--GAIN	dB1	[R/S]	DIA=
5.	To compute 3dB beamwidth		[XEQ] 3db	FREQ.?
	--input frequency (GHz)	f	[R/S]	% EFF.?
	--% Efficiency	% E	[R/S]	DIA. ?
	--antenna diameter	DIA	[R/S]	$\measuredangle$=
	--to compute ndB (below max)		[XEQ] NdB	N?
	--input N	N	[R/S]	$\measuredangle$ =
6.	To compute dB below max		[XEQ] dB	FREQ.?
	--input frequency (GHz)	f	[R/S]	% EFF.?
	--% EFFICIENCY	%E	[R/S]	$\measuredangle$?
	--off-axis angle (deg)		[R/S]	dB=
*	Note: the frequency and % efficiency will			
	not be asked for after one run.			

Program Listings

```
01◆LBL a
02 CF 00
03 "FREQ.?"
04 PROMPT
05 STO 02
06 LOG
07 20
08 STO 06          20 Log f → R03
09 *
10 STO 03
11 3.28
12 STO 04
13 "% EFF.?
"
14 PROMPT
15 ENTER↑
16 .102
17 *
18 LOG
19 10              Eff. constant
20 *
21 ST+ 03
22 RTN
23◆LBL "GAI
N"                 GAIN
24 FS? 00          Initialize
25 XEQ a
26 "DIA. ?"
27 PROMPT
28 FS? 01
29 XEQ 09
30 LOG
31 RCL 06
32 *
33 RCL 03
34 +
35 "dB1="          Display dB1
36 ARCL X
37 AVIEW
38 STOP
39◆LBL "DIA       DIA
"
40 FS? 00          Initialize
41 XEQ a
42 "dB1?"
43 PROMPT
44 RCL 03
45 -
46 RCL 06
47 /
48 10↑X
49 FS? 01
50 XEQ 08
51 "DIA="          display DIA
52 ARCL X
53 AVIEW
54 STOP
55◆LBL 08
56 RCL 04
57 /
58 RTN
59◆LBL 09
60 RCL 04
61 *
62 RTN
63◆LBL "3dB       3dB
"
64 FS? 00
65 XEQ a
66 "DIA. ?"
67 PROMPT
68 FS? 01
69 XEQ 09
70 RCL 02
71 *
72 66.4
73 X<>Y
74 /
75 STO 01
76 "∡="            Display
77 ARCL X
78 AVIEW
79 STOP
80◆LBL "NdB       NdB
"
81 FS? 00
82 XEQ a
83 "N?"
84 PROMPT
85 RCL 01
86 X↑2
87 *
88 3
89 /
90 SQRT
91 "∡="            Display
92 ARCL X
93 AVIEW
94 STOP
95◆LBL "dB"      dB
96 FS? 00
97 XEQ a
```

Program Listings

```
 98 "∡?"
 99 PROMPT
100 X↑2
101 12
102 *
103 RCL 01
104 X↑2
105 /
106 "dB="
107 ARCL X          Display dB
108 AVIEW
109 STOP
110 .END.
```

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

#		#	
00		50	
	3dB BW		
	f (GHz)		
	20 LOG f		
	meter/feet conv.		
05		55	
	20		
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 007 TOT. REG. 40 USER MODE

ENG ___ FIX 2 SCI ___ ON ___ OFF

DEG X RAD ___ GRAD ___

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
00	S	have f + % Eff.	need f + % eff.
01	*	meter	feet
	*	Set or clear as desired.	

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY
None			

RF PATH LOSS, dB

Using the following equations:

$$L_1 = 20 \log (41.87\ fD)$$

$$L_2 = 66.2 + 1070\left(\frac{H}{D}\right) - 7500\left(\frac{H}{D}\right)^2 + 0.00268\ f + 28.34 \log f + 0.879\ D - 0.00378\ D^2$$

$$L_3 = -10 \log \left\{ \frac{1.033 \times 10^{-3}}{(fD)^4} \left[1 + 31.1(fH_1)^2\right] \left[1 + 31.1(fH_2)^2\right] \right\}$$

the RF path loss in dB is calculated.

Where: L_1 is the free space path loss

L_2 is the path loss due to an obstacle in the path

L_3 is the smooth earth path loss

Note: Output data valid only for VHF; ie , 20 MHz < f < 500 MHz.

Example:

What is the free space RF path loss and the RF path loss due to a terrain obstacle for the following conditions:

RF frequency	20 MHz	(f)
Distance between antennas	5 Km	(D)
Height of antenna 1	0.1 Km	(H_1)
Height of antenna 2	0.01 Km	(H_2)
Height of obstacle above line-of-sight between the antennas	0.01 Km	(H)

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 008	
[XEQ] [ALPHA] PL [ALPHA]	FREQ.?
20 [R/S]	D?
5 [R/S]	H1?
.1 [R/S]	H2?
.01 [R/S]	H?
.01 [R/S]	L1=72.44
[R/S]	L2=109.54
[R/S]	L3=85.38

USER INSTRUCTIONS

SIZE: 008

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load Program			
2	Begin execution and input:		[XEQ] PL	FREQ.?
	--frequency (MHz)	f	[R/S]	D?
	--distance between antennas (Km)	D	[R/S]	H1?
	--height of antenna 1 (Km)	H_1	[R/S]	H2?
	--height of antenna 2 (Km)	H_2	[R/S]	H?
	--height of obstacle (above line-of-sight)			
	between antennas	H	[R/S]	L1=
			[R/S]	L2=
			[R/S]	L3=

Program Listings

Program	Comments	Program	Comments
01◆LBL "PL"	Initialization	52 *	
02 FIX 2		53 +	
03 "FREQ.?"		54 .88	
04 PROMPT		55 RCL 06	
05 STO 05		56 *	
06 "D?"		57 +	
07 PROMPT		58 3.78 E-3	
08 STO 06		59 RCL 06	
09 "H1?"		60 X↑2	
10 PROMPT		61 *	
11 STO 02		62 −	
12 "H2?"		63 "L2="	Display L_2
13 PROMPT		64 XEQ d	
14 STO 03		65 1.03 E-3	
15 "H?"		66 RCL 05	
16 PROMPT		67 RCL 06	
17 STO 04		68 *	
18 20		69 X↑2	
19 STO 07		70 X↑2	
20 41.87		71 /	
21 RCL 05		72 31.1	
22 RCL 06		73 RCL 05	
23 *		74 RCL 02	
24 *		75 *	
25 LOG		76 X↑2	
26 RCL 07		77 *	
27 *		78 1	
28 "L1="	Display L_1	79 +	
29 XEQ d		80 *	
30 66.2		81 31.1	
31 ENTER↑		82 RCL 05	
32 1070		83 RCL 03	
33 RCL 04		84 *	
34 RCL 06		85 X↑2	
35 /		86 *	
36 *		87 1	
37 +		88 +	
38 7500		89 *	
39 RCL 04		90 LOG	
40 RCL 06		91 RCL 07	
41 /		92 2	
42 X↑2		93 /	
43 *		94 *	
44 −		95 CHS	
45 2.68 E-3		96 "L3="	Display L_3
46 RCL 05		97◆LBL d	
47 *		98 ARCL X	
48 +		99 AVIEW	
49 28.34		100 STOP	
50 RCL 05		101 .END.	
51 LOG			

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00		50	
	H_1		
	H_2		
	H		
05	f	55	
	D		
	Temp. storage		
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 008 TOT. REG. 32 USER MODE
ENG ___ FIX 2 SCI ___ ON ___ OFF X
DEG ___ RAD ___ GRAD ___

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

ANTENNA GAIN OR POWER OF A REMOTE TRANSMITTER

Program computes T' using the formula:

$$T' = \frac{(4\pi)^2 \; f^2 \; R^2 \; P_R}{T \; G_R \; C^2}$$

Where

f = frequency of RF carrier (MHz)

R = range from xmitter or rcvr (Km)

P_R = strength of received signal (watts)

G_R = gain of rcvr antenna (dB)

T = gain of xmit antenna (or xmit power) (dB)

T' = xmit power (or gain or xmit antenna) (watts)

and C = speed of light (0.29979250 km/μs)

Note: Values are ideal; user should apply any known system losses to avoid error in computation (propagation, etc.)

Example:

Given rcvr antenna gain of 15 dB; transmitter located 750 statute miles away; frequency of 120 MHz; 15 x 10^{-6} watts received signal; and transmit antenna gain of 300 dB, compute transmitted power.

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 001	
[XEQ] [ALPHA] AG [ALPHA]	GR?
15 [R/S]	R?
750 [ENTER] 1.609347219 [x] [R/S]	FREQ.?
120 [R/S]	POWER RECVD?
15 [EEX] [CHS] 6 [R/S]	POWER OR GAIN?
300 [R/S]	122,868.76

User Instructions

SIZE: 001

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1.	Load Program			
2.	Begin execution and input:		[XEQ] AG	GR?
	--rcvr antenna gain	G_R	[R/S]	R?
	--range	R	[R/S]	FREQ.?
	--frequency of RF carrier	f	[R/S]	POWER RECVD?
	--strength of received signal	P_R	[R/S]	POWER OR GAIN?
3.	To compute antenna gain:			
	--input transmitted power	T'	[R/S]	(T)
4.	To compute transmitted power:			
	--input antenna gain	T	[R/S]	(T')
5.	If no change to G_R, R, f; push [R/S] for			
	new case		[R/S]	POWER RECVD?

Program Listings

`01◆LBL "AG"`	Initialize
`02 PI`	
`03 4`	
`04 *`	
`05 ENTER↑`	
`06 *`	
`07 .2997925`	
`08 ENTER↑`	
`09 *`	
`10 /`	
`11 STO 00`	
`12 CLX`	
`13 "GR?"`	
`14 PROMPT`	
`15 ST/ 00`	
`16 "R?"`	
`17 PROMPT`	
`18 "FREQ.?"`	
`19 PROMPT`	
`20 *`	
`21 ENTER↑`	
`22 *`	
`23 ST* 00`	
`24◆LBL A`	
`25 "POWER RECVD?"`	
`26 PROMPT`	
`27 "POWER OR GAIN?"`	
`28 PROMPT`	
`29 /`	
`30 RCL 00`	
`31 *`	
`32 RTN`	Display Power or Gain
`33 GTO A`	
`34 .END.`	

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	Temp storage	50	
05		55	
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 001 TOT. REG. 13 USER MODE
ENG ___ FIX 2 SCI ___ ON ___ OFF X
DEG ___ RAD ___ GRAD ___

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
		None	

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY
None			

PLANAR PHASED ARRAY RADAR BEAM POSITIONS

Coordinate conversion between boresight plane and any other rotated plane by:

1) Converting spherical coordinates ($\sin\alpha$, $\sin\beta$, R=unity) to rectangular coordinates where $x = \sin\theta\cos\phi$; $y = \sin\theta\sin\phi$, $z = \cos\theta$.
2) Rotating the boresight plane to the new plane and computing new rectangular coordinates using the "directed angle cosines".
3) Converting the new rectangular coordinates to spherical coordinates in the new plane where $\theta = \sin^{-1}z$, $\phi = \tan^{-1}x/y$.

The inverse of the above procedure using angle inputs and obtaining $\sin\alpha$, $\sin\beta$ outputs is also used. The above steps are accomplished using the following formulas:

A. Conversion of $\sin\alpha$, $\sin\beta$ to α', β' where BS = Boresight angle

$$\alpha' = \mathrm{Tan}^{-1}\left[\frac{\sin\alpha\cos\beta}{\cos BS\cos\alpha\cos\beta - \sin BS\sin\beta}\right]$$

$$\beta' = \sin^{-1}\left[\sin BS\cos\alpha\cos\beta + \cos BS\sin\beta\right]$$

B. Conversion of α, β to $\sin\alpha'$, $\sin\beta'$, where BS = Boresight angle

$$\sin\alpha' = \sin\left(\mathrm{Tan}^{-1}\left(\frac{\sin\alpha\cos\beta}{\cos BS\cos\alpha\cos\beta + \sin BS\sin\beta}\right)\right)$$

$$\sin\beta' = (\cos BS\sin\beta - \sin BS\cos\alpha\cos\beta)$$

Radar boresight (BS) elevation planes can be used within the following limits $0° \leq BS° \leq 90°$. For sign convention, elevations below boresight should be entered as a negative. Azimuth sines to the left of boresight (viewers eyes looking out of boresight) should be entered as negatives. All other sines should be entered as positive.





Example:

Using a boresight angle of 45° with the earth plane, convert the following sin α and sin β beam positions to spherical coordinates (dec. deg.) (azimuth and elevation) in the tangent earth plane:

a. $\sin\alpha = -.7071$, $\sin\beta = -.3$;

b. $\sin\alpha = .707$, $\sin\beta = .3$;

c. $\sin\alpha = .32$, $\sin\beta = .6$.

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 010	
[XEQ] [ALPHA] DEG [ALPHA]	BS?
45 [R/S]	SIN ALPHA?
.7071 [CHS] [R/S]	SIN BETA?
.3 [CHS] [R/S]	AZ=-44.39 DEG
[R/S]	ELE.=15.36 DEG
[R/S]	SIN ALPHA?
.707 [R/S]	SIN BETA?
.3 [R/S]	AZ=68.56 DEG
[R/S]	ELE.=43.56 DEG
[R/S]	SIN ALPHA?
.32 [R/S]	SIN BETA?
.6 [R/S]	AZ=66.43 DEG
[R/S]	ELE.=73.78 DEG

User Instructions

SIZE: 010

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Begin execution			
2	To compute azimuth position and elevation			
	position in degrees		[XEQ]DEG	BS?
	--input boresight elevation	BS	[R/S]	SIN ALPHA?
	--sin α	sin α	[R/S]	SIN BETA?
	--sin β	sin β	[R/S]	AZ =( )DEG
			[R/S]	ELE.=( )DEG
3	To compute sin α, sin β		[XEQ] SINE	BS?
	--input boresight elevation	BS	[R/S]	DEG AZ?
	--azimuth	Dec. Deg.	[R/S]	DEG ELE.?
	--elevation	Dec. Deg.	[R/S]	SIN ALPHA=
			[R/S]	SIN BETA=
4	Step 2 or 3 may be repeated without the			
	BS prompt by pushing [R/S]			

Program Listings

Program	Comment
01♦LBL "DEG"	DEG
02 FIX 2	
03 "BS?"	
04 PROMPT	Initialization
05 STO 07	
06♦LBL A	
07 "SIN ALPHA?"	
08 PROMPT	
09 "SIN BETA?"	
10 PROMPT	
11 STO 01	
12 ASIN	
13 COS	
14 STO 02	
15 X<>Y	
16 STO 03	
17 ASIN	
18 COS	
19 STO 04	
20 XEQ C	
21 *	
22 X<>Y	
23 RCL 01	
24 *	
25 -	
26 XEQ D	
27 STO 08	
28 "AZ="	display AZ
29 XEQ d	
30 XEQ E	
31 +	
32 ASIN	
33 "ELE.="	display ELE
34 XEQ d	
35 GTO A	
36♦LBL d	
37 ARCL X	
38 "├ DEG"	display
39 AVIEW	
40 STOP	
41 RTN	
42♦LBL "SINE"	SINE
43 FIX 2	
44 "BS?"	
45 PROMPT	
46 STO 07	
47♦LBL B	
48 "DEG AZ?"	Initialization
49 PROMPT	
50 "DEG ELE.?"	
51 PROMPT	
52 ENTER↑	
53 ENTER↑	
54 1	
55 *	
56 SIN	
57 STO 01	
58 RDN	
59 COS	
60 STO 02	
61 RDN	
62 SIN	
63 STO 03	
64 RDN	
65 COS	
66 STO 04	
67 XEQ C	
68 RCL 02	
69 *	
70 R↑	
71 RCL 01	
72 *	
73 +	
74 XEQ D	
75 SIN	
76 STO 08	
77 "SIN ALPHA="	display SIN ALPHA
78 XEQ b	
79 XEQ E	
80 -	
81 "SIN BETA="	display SIN BETA
82 XEQ b	
83 GTO B	
84♦LBL b	
85 ARCL X	
86 AVIEW	
87 STOP	
88 RTN	
89♦LBL C	Common subroutines
90 RCL 07	
91 COS	
92 STO 05	
93 RCL 07	
94 SIN	

Program Listings

```
 95 STO 06
 96 RDN
 97 *
 98 RTN
 99+LBL D
100 RCL 02
101 RCL 03
102 *
103 X<>Y
104 /
105 ATAN
106 RTN
107+LBL E
108 RCL 05
109 RCL 01
110 *
111 RCL 06
112 RCL 04
113 RCL 02
114 *
115 *
116 RTN
117 .END.
```

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00		50	
	sin β		
	cos β		
	sin α		
	cos α		
05	cos BS	55	
	sin BS		
	BS (deg)		
	Deg (Az)		
	Temp storage		
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE .010 TOT. REG. 44 USER MODE
ENG ___ FIX 2 SCI ___ ON ___ OFF X
DEG X RAD ___ GRAD ___

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
		None	

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY
None			

SHORT WAVE TRANSMISSION PATH CALCULATIONS

This program calculates parameters that are of particular interest to ham operators and short wave listeners:

1. Bearing angles from true north between transmitter and receiver (A_{TR}), and receiver and transmitter (A_{RT}).

2. Distance between the two points in miles (D_{mi}) and kilometers (D_{km}), and number of hops (N). Maximum length of one hop is 36 or 4000 km.

3. Coordinates for reflection points:

   Reflection points are counted from the transmitter towards the receiver:

m=1	TRCP	Transmitter reflection control point
m=2N-1	RRCP	Receiver reflection control point
m=0		Transmitter location
m=2N		Receiver location

   Odd values (m-1, 3, etc.) for ionespheric reflection points
   Even values (m=2, 4, etc.) for ground reflection points

4. The "most northerly point" (MNP) on the transmission path.

This program requires an extra memory module.

Equations:



$$\cos d = \sin L_T \sin L_R + \cos L_T \cos L_R \cos (\lambda_R - \lambda_T)$$

$$\cos R = \frac{\sin L_T - \cos d \sin L_R}{\sin d \cos L_R}$$

$$\cos T = \frac{\sin L_R - \cos d \sin L_T}{\sin d \cos L_T}$$

$$A_{TR} = \begin{cases} T \quad , \text{ if } -180 < \lambda_R - \lambda_T < 0 \\ 360-T, \text{ if } \quad 0 \le \lambda_R - \lambda_T \le 180 \end{cases}$$

$$A_{RT} = \begin{cases} 360-R, \text{ if } -180 < \lambda_R - \lambda_T < 0 \\ R \quad , \text{ if } \quad 0 \le \lambda_R - \lambda_T \le 180 \end{cases}$$

$$N = \text{INT}\left[\frac{d}{36} + 1\right]$$

$$D_{mi} = 69.05\ d$$

$$D_{km} = 111.12\ d$$





$$C_{RP} = \left(1 - \frac{m}{2N}\right) d$$

$$C_{MNP} = \tan^{-1} \frac{\cos R}{\tan L_R}$$

$$\sin L_C = \cos C \sin L_R + \sin C \cos L_R \cos R$$

$$\Delta\lambda = \Delta\lambda' \frac{|\lambda_T - \lambda_R|}{(\lambda_T - \lambda_R)}$$

$$\lambda_C = \lambda_R + \Delta\lambda$$

$$\cos \Delta\lambda' = \frac{\cos C - \sin L_C \sin L_R}{\cos L_C \cos L_R}$$

where:

- NP = north pole
- λ_R = receiver longitude
- λ_T = transmitter longitude
- λ_C = longitude of point along path
- L_R = receiver latitude
- L_T = transmitter latitude
- L_C = latitude of point along path
- B_{RT} = bearing to transmitter
- B_{TR} = bearing to receiver
- d = angle from transmitter to receiver
- N = number of hops
- D_{mi} = distance from transmitter to receiver in miles
- D_{Km} = distance from transmitter to receiver in kilometers
- C_{RP} = angle from receiver to reflection point
- C_{MNP} = angle from receiver to most northerly point
- MNP = the most northerly point between the receiver and transmitter
- m = an integer representing the mth reflection point

Remarks:

This program calculates the parameters for the short path transmission only ($d \leq 180$).

For either transmitter or receiver location at the north or south pole ($L=90$), the program will display "DATA ERROR."

Longitudes (λ) are taken positive for points west, negative for points east of the Greenwich Meridian.

The latitudes (L_e) of a point on the northern hemisphere is positive, on the southern hemisphere, negative.

All coordinates are entered and displayed in degrees, minutes,and seconds.

Reference:

This program is a translation of the HP67/97 program 00695D from the User's Library.

Example 1:



R is in south San Francisco, California, USA. Lat. = 37°40'N; Long. = 122°24'W

T is in Johannesburg, SA. Lat. = 26°35'S; Long. = 28°8'E

Find the two bearings, the number of hops, the distance in miles and kilometers, the two reflection control points, the most northerly point, and the 5th reflection point.

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 013	
[XEQ] [ALPHA] SWTP [ALPHA]	REC. LAT. ?
37.4 [R/S]	REC. LONG. ?
122.24 [R/S]	TRAN. LAT. ?
26.35 [CHS] [R/S]	TRAN. LONG. ?
28.8 [CHS] [R/S]	BTR=302.0131 (302°1'31'')
[R/S]	BRT=73.1754 (73°17'54'')
[XEQ] [ALPHA] N [ALPHA]	N=5
[R/S]	DMI=10,616.99 (miles)
[R/S]	DKM=17,085.58 (Km)
[XEQ] [ALPHA] RCP [ALPHA]	RLAT.=40.3018 (RRCP)
[R/S]	RLONG.=102.5316
[R/S]	TLAT.=-17.4810 (TRCP)
[R/S]	TLONG.=-15.4036
[XEQ] [ALPHA] MNP [ALPHA]	NLAT.=40.4144 (MNP)
[R/S]	NLONG.=96.1454
[XEQ] [ALPHA] MRP [ALPHA]	M?
5 [R/S]	IONESPHERE
[R/S]	MLAT.=21.0660
[R/S]	MLONG.=32.5557

Example 2:



R is in South San Francisco, California, USA. Lat. = 37°40'N; Long. = 122°24'W

T is in Shepparton, Victoria, AUSTR. Lat. = 36°20'S; Long. = 145°25'E

Find the bearings from transmitter to receiver and receiver to transmitter. Also find the most northerly point.

Keystrokes:	Display:
[XEQ] [ALPHA] SWTP [ALPHA]	REC. LAT. ?
37.4 [R/S]	REC. LONG. ?
122.24 [R/S]	TRAN. LAT. ?
36.2 [CHS] [R/S]	TRAN. LONG. ?
145.25 [CHS] [R/S]	BTR=59.0250 (59°2'50'')
[R/S]	BRT=240.4653 (240°46'53'')
[XEQ] [ALPHA] MNP [ALPHA]	NO MNP

(There is no MNP between the transmitter and the receiver.)

User Instructions

SIZE: 013

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
	Load program			
2	Initialize		[XEQ] SWTP	REC. LAT.?
	Key in the receiver's latitude	L_r	[R/S]	REC. LONG.?
	Key in the receiver's longitude	λ_r	[R/S]	TRAN. LAT.?
	Key in the trnasmitter's latitude	L_t	[R/S]	TRAN. LONG.?
	Key in the transmitter's longitude	λ_t	[R/S]	BTR=(D.MS)
	Calculated are the bearings from trans-			
	mitter to receiver and receiver to trans-			
	mitter.		[R/S]	BRT=(D.MS)
	Go to steps 3, 4, 5, or 6 as desired			
3	Calculate the number of hops,		[XEQ] N	N=
	distance in miles,		[R/S]	DMI=(miles)
	and kilometers.		[R/S]	DKM=(km)
	Calculate the reflection control		[XEQ] RCP	RLAT=(D.MS)
	points for receiver		[R/S]	RLONG.=(D.MS)
	and transmitter		[R/S]	TLAT.=(D.MS)
			[R/S]	TLONG.=(D.MS)
5	Calculate the most northerly point		[XEQ] MNP	LAT.=(D.MS)
			[R/S]	LONG.=(D.MS)
6	Calculate the coordinates of any		[XEQ] MRP	M?
	m th reflection point	m	[R/S]	MLAT.=
			[R/S]	MLONG.=
			[R/S]	IONESPHERE (or) GROUND

Program Listings

Initialization

```
01♦LBL "SWT
P"
02 FIX 4
03 DEG
04 "LAT."
05 ASTO 10
06 "LONG."
07 ASTO 11
08 "REC. "
09 ASTO 12
10 ARCL 10
11 "⊢ ?"
12 PROMPT
13 HR
14 STO 02
15 CLA
16 ARCL 12
17 ARCL 11
18 "⊢ ?"
19 PROMPT
20 HR
21 STO 01
22 "TRAN. "
23 ASTO 12
24 ARCL 10
25 "⊢ ?"
26 PROMPT
27 HR
28 STO 05
29 CLA
30 ARCL 12
31 ARCL 11
32 "⊢ ?"
33 PROMPT
34 HR
35 STO 04
```

Calculate d

```
36 RCL 01
37 -
38 COS
39 RCL 02
40 COS
41 *
42 RCL 05
43 COS
44 *
45 RCL 02
46 SIN
47 RCL 05
48 SIN
49 *
50 +
51 ACOS
52 STO 00
```

Calculate R and T

```
53 RCL 02
54 RCL 05
55 XEQ 02
56 STO 03
57 RCL 05
58 RCL 02
59 XEQ 02
```

Calculate N

```
60 STO 06
61 RCL 00
62 36
63 /
64 1
65 +
66 INT
67 STO 07
```

Calculate B_{TR} and B_{RT}

```
68 RCL 04
69 RCL 01
70 -
71 1
72 P-R
73 R-P
74 RDN
75 X>0?
```

R and (360-T)

```
76 GTO 01
77 RCL 03
78 HMS
79 360
80 RCL 06
81 -
```

(360-R) and T

```
82 GTO 05
83♦LBL 01
84 360
85 RCL 03
86 -
87 HMS
```

Display B_{TR} and B_{RT}

```
88 RCL 06
89♦LBL 05
90 HMS
91 "BTR="
92 ARCL X
93 PROMPT
94 "BRT="
95 ARCL Y
96 PROMPT
```

Subroutine to calculate either R or T

```
97 RTN
98♦LBL 02
99 SIN
100 X<>Y
101 STO 12
```

Program Listings

Program	Comment
102 SIN	
103 RCL 00	
104 COS	
105 *	
106 -	
107 RCL 00	
108 SIN	
109 RCL 12	
110 COS	
111 *	
112 /	
113 ACOS	
114 RTN	
115◆LBL "N"	Display N
116 FIX 0	
117 CF 29	
118 "N="	
119 ARCL 07	
120 FIX 2	
121 SF 29	
122 PROMPT	
123 RCL 00	Display D_{mi}
124 69.05	
125 *	
126 "DMI="	
127 ARCL X	
128 PROMPT	
129 RCL 00	Display D_{km}
130 111.12	
131 *	
132 "DKM="	
133 ARCL X	
134 PROMPT	
135 RTN	Calculate re-
136◆LBL "RCP"	flection control
	points
137 RCL 07	
138 2	
139 *	
140 1	
141 -	
142 "R"	m_{RRCP}
143 ASTO 08	
144 XEQ 09	
145 1	
146 "T"	
147 ASTO 08	
148 GTO 09	Find coordinates
149◆LBL "MNP"	m_{TRCP}
150 90	Calculate MNP
151 RCL 03	
152 X≠Y?	
153 X>Y?	
154 GTO 06	If 90° ≤ R then
155 COS	no MNP
156 RCL 02	
157 TAN	
158 /	
159 ATAN	
160 PI	
161 R-D	
162 X<>Y	
163 X<0?	
164 +	
165 "N"	
166 ASTO 08	
167 RCL 00	
168 X<>Y	
169 X<=Y?	If c > d then no
170 GTO 03	MNP
171◆LBL 06	find coordinates
172 "NO MNP"	
173 PROMPT	
174 RTN	Calculate mth
175◆LBL "MRP"	reflection point
176 "M?"	
177 PROMPT	
178 "M"	
179 ASTO 08	
180 STO 12	
181 2	
182 /	
183 FRC	
184 "IONESPHERE"	
185 X=0?	
186 "GROUND"	
187 PROMPT	
188 RCL 12	
189◆LBL 09	convert m to c
190 RCL 07	
191 2	
192 *	
193 /	
194 CHS	
195 1	
196 +	
197 RCL 00	take c and calcu-
198 *	late coordinates
199◆LBL 03	

Program Listings

```
FIX 4
STO 09
COS
RCL 02
SIN
*
RCL 09
SIN
RCL 02
COS
*
RCL 03
COS
*
+
ASIN          Lc
STO 12
HMS
CLA
ARCL 08
ARCL 10
"⊢="
ARCL X
PROMPT        Display Lc
RCL 09
COS
RCL 12
SIN
RCL 02
SIN
*
-
RCL 12
COS
/
RCL 02
COS
/
ACOS          Δλ'
RCL 04
RCL 01
-
1
P-R
R-P
RDN
ABS
X=0?
GTO 04
LASTX
/
```

```
*
◆LBL 04       Δλ
RCL 01
+
1
P-R
R-P
RDN           λc
HMS
CLA
ARCL 08
ARCL 11
"⊢="
ARCL X
PROMPT        Display λc
RTN
.END.
```

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	d	50	
	λ_R		
	L_R		
	R		
	λ_T		
5	L_T	55	
	T		
	N		
	Alpha		
	Temp		
10	Alpha	60	
	Alpha		
	Temp		
5		65	
20		70	
5		75	
30		80	
5		85	
40		90	
5		95	

STATUS

SIZE 013. TOT. REG. 75 USER MODE

ENG ___ FIX ___ SCI ___ ON ___ OFF X

DEG ___ RAD ___ GRAD ___

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

Hewlett-Packard Software

In terms of power and flexibility, the problem-solving potential of the HP-41 programmable calculator is nearly limitless. And in order to see the practical side of this potential, HP has different types of software to help save you time and programming effort. Every one of our software solutions has been carefully selected to effectively increase your problem-solving potential. Chances are, we already have the solutions you're looking for.

Application Pacs

To increase the versatility of your HP-41, HP has an extensive library of "Application Pacs". These programs transform your HP-41 into a specialized calculator in seconds. Included in these pacs are detailed manuals with examples, miniature plug-in Application Modules, and keyboard overlays. Every Application Pac has been designed to extend the capabilities of the HP-41.

You can choose from:

Aviation (Pre-Flight Only) 00041-15018
Clinical Lab 00041-15024
Circuit Analysis 00041-15024
Financial Decisions 00041-15004
Mathematics 00041-15003
Structural Analysis 00041-15021
Surveying 00041-15005
Securities 00041-15026
Statistics 00041-15002
Stress Analysis 00041-15027
Games 00041-15022
Home Management 00041-15023
Machine Design 00041-15020
Navigation 00041-15017
Real Estate 00041-15016
Thermal and Transport Science 00041-15019
***Petroleum Fluids 00041-15039**

Users' Library

The Users' Library provides the best programs from contributors and makes them available to you. By subscribing to the HP-41 Users' Library you'll have at your fingertips literally hundreds of different programs from many different application areas.

*Users' Library Solutions Books

Hewlett-Packard offers a wide selection of Solutions Books complete with user instructions, examples, and listings. These solution books will complement our other software offerings and provide you with a valuable tool for program solutions.

You can choose from:

Business Stat/Marketing/Sales 00041-90094
Home Construction Estimating 00041-90096
Lending, Saving and Leasing 00041-90086
Real Estate 00041-90136
Small Business 00041-90137
Geometry 00041-90084
High-Level Math 00041-90083
Test Statistics 00041-90082
Antennas 00041-90093
Chemical Engineering 00041-90100
Control Systems 00041-90092
Electrical Engineering 00041-90088
Fluid Dynamics and Hydraulics 00041-90139
Games II 00041-90443
Civil Engineering 00041-90089
Heating, Ventilating & Air Conditioning 00041-90140
Mechanical Engineering 00041-90090
Solar Engineering 00041-90138
Calendars 00041-90145
Cardiac/Pulmonary 00041-90097
Chemistry 00041-90102
Games 00041-90099
Optometry I (General) 00041-90143
Optometry II (Contact Lens) 00041-90144
Physics 00041-90142
Surveying 00041-90141
Time Module Solutions 00041-90395

*Some books require additional memory modules to accomodate all programs.